

PXTRANSFER:

An MBASIC File Transfer Program

by Joseph A. Sabin, Jr.

PX-Transfer is a semi-intelligent file transfer system. The reason I call it semi-intelligent is because it was designed to overcome some of the problems in sending and receiving text files with mainframe computer systems that do not have a file transfer protocol.

The name PX-Transfer comes from PXEDIT, which is the name given to the Princeton University version of IBM's XEDIT program. PX-Transfer has been designed for Microsoft BASIC in either its compiled or interpreted form. To make it work with the interpreted BASIC, line number 1041 needs to be eliminated. Loops occur approximately 20 times as fast in the compiled versions as they do in the interpreter. Line number 1041 modifies the value of MHZ for this purpose. It is possible that this value may need adapting for your computer system.

Interpreted, PX-Transfer works fine at 300 baud with a 2.5 Mhz Z80 system. Compiled, I have had success at 1200 baud. The code as it is presented will work on a Kaypro 2 computer, and a Hayes Smartmodem. The Mhz value needs to be changed to 4.00 for the Kaypro 4 or 10. If you have a different computer you will have to change the data statement values to those matching your computer.

In as much of the code as possible I have included comments to explain what it is you will need to change. If you will be communicating to an IBM 3081 and XEDIT it should just work. On the other hand, if not, you will need to change the areas marked in the code to the correct values. At first I tried to put all the things in an easy-to-find area in the code, but it was too slow. Therefore I designed the code in modules to handle the individual areas of activity.

Much of the system-dependent information is contained in data statements at the beginning of the program. These can be easily changed in either the interpreted or compiled form. The Microsoft compiler leaves data statements in their ASCII format with commas between the values, just as they appear in the source code. The information contained in the data statements is as follows:

Description

- 1: This is the data port number
- 2: This is the status port number
- 3: The value to AND the status port input value with to see if there is an incoming character.
- 4: The value to AND the status port input value with to see if the outgoing buffer is empty.
- 5: Clock speed in Mhz.
- 6: Height of the screen less 1
- 7: Width of the screen less 1
- 8: Baud rate port number
- 9: Number of characters to set Baud rate
- 10-
- 11: Baud rate characters (2)
- 12-
- 15: Clear screen characters (4)
- 16: String containing the name of the computer the system is set up for.

PX-Transfer has sufficient capability as a dumb terminal program to allow logging on the user's computer to the mainframe. The purpose of this program is to allow the user to send and receive text files with a computer without a communications system accessible. Since most mainframe computers have a line editor that sends and receives text line by line (such as XEDIT by IBM) this program thus controls the transfer of text by using the commands of the editor program.

The concept behind this program is that XEDIT (or whatever) can be told to print a line of text by command to the user. This line of text is captured by PX-Transfer and then saved to the disk file. Once that line is saved PX-Transfer requests the next line in the file. This process continues until the EOF statement is received by PX-Transfer.

To make PX-Transfer work with the mainframe computer you want to communicate to, you need to know what the command structure is for the line editor. Second, you need to know the prompt sent by the mainframe to the terminal (your microcomputer). Prompts usually consist of several characters. To be safe, I use the second to last character in the prompt to signal the editor's readiness to either send or receive the next line of text. The last character of the prompt is then avoided by use of the timing loop.

The user interface is big enough to let you know what is going on without getting in the way of the communication. One problem I have encountered in using the program as just a dumb terminal is that BASIC string garbage collection occasionally causes a series of dropped characters. This problem is virtually nonexistent in the smart section because all the incoming data is kept in strings of 128 characters each, and therefore little garbage collection goes on.

If you use the AUTO function in MBASIC to input the program you will do just fine, but will not get many of the comments in the code included. This is of course fine for your own use, but if this is to be given away in source please include all the comments. If comments are deleted it will be difficult to adapt to a new computer system.

The code includes a status value for I/O port out clear, but never uses it. This is there for expansion at a future time. Instead the code uses a timing loop to make sure that the characters are sent at about 22-26 characters per second. At 300 baud I have found this to be the optimum speed for the transfer. If the letters go any faster it seems as if the big dumb machine at the other end gets "mad" and hangs up. So instead we make it "think" we are very fast typists (300 words per minute with no falter in our stride).

This is not a communications program; rather, it is an easy way to send and receive text from an otherwise reluctant host computer. Do give it a try, especially if there is a need in your area for such a program.

Permission is granted to adapt this program to the local computer and then distribute it locally. The only stipulation is that my copyright remains in the code and you support your local version of the program.

```

999 '↓----- to be changed -----↓'
1000 DEFINT A,B:DATA 004,006,001,004,2.50,23,79,000,0,000,000
1010 DATA 000,000,000,026,'Kaypro II '
1011 '↑-----↑'
1020 READ DTA,STAT,OKI,OKO,MHZ,SH,SW,BRP,NBRP,BRP(1),BRP(2)
1030 FOR I = 1 TO 4:READ A:CLS$ = CLS$ + CHR$(A):NEXT
1040 READ COMP$
1041 MHZ = MHZ*20:'*** this is only for the compiled Version ***
1050 FOR I = 1 TO NBRP:OUT BRP,BRP(I):TIME1 = MHZ:GOSUB 3330:NEXT
1060 IF RIGHT$(COMP$,1) = " " THEN COMP$ = LEFT$(COMP$,LEN(COMP$) - 1):GOTO 1060
1069 '**** initially set up for half duplex — take care with this if you change the default ****
1070 DIM A$(5):LL = SW:DUP$ = 'Half Duplex type Control R for Full':DUP = 1
1080 GOSUB 1090:GOTO 1220
1089 ' **** Startup message and Control I prompt ****
1090 PRINT CLS$:
1100 PRINT:PRINT SPC(sw/2 - len(comp$)/2);comp$
1110 PRINT TAB(12); ***** PX-Transfer Version 1.01 *****
1120 PRINT TAB(12); "
1130 PRINT TAB(12); " CONTROL A FOR AUTODIAL
1140 PRINT TAB(12); " CONTROL C TO CLEAR THE SCREEN
1150 PRINT TAB(12); " CONTROL D TO DISCONNECT MODEM
1160 PRINT TAB(12); " CONTROL I FOR THESE INSTRUCTIONS
1170 PRINT TAB(12); " CONTROL T TO TRANSFER A FILE
1180 PRINT TAB(12); " CONTROL X TO EXIT THE SYSTEM
1190 PRINT TAB(12); "
1200 PRINT TAB(12); ***** (c) 1983 MicroFrame Associates *****
1201 *****↑----- must be here -----↑*****
1210 RETURN
1220 PRINT:PRINT DUP$
1230 PRINT CHR$(13);:PRINT'Ready →';
1240 A$ = INKEY$:IF (INP(STAT) AND OKI) = OKI THEN B = (INP(DTA) AND &H7F):IF (B>31 OR B = 13 OR B = 10 OR B = 8) AND B<>127 THEN PRINT
CHR$(B);
1250 IF A$ = " " THEN 1240 ELSE A = ASC(A$):IF (A>31 OR A = 13 OR A = 8 OR A = 27)THEN OUT DTA,A:IF DUP = 1 THEN PRINT A$;:GOTO
1240:ELSE GOTO 1240
1260 IF A = 1 THEN GOSUB 1340:PRINT:PRINT"Waiting . . .":GOTO 1240
1270 IF A = 3 THEN PRINT CLS$;:GOTO 1240
1280 IF A = 4 THEN GOSUB 2180:GOTO 1220
1290 IF A = 9 THEN GOSUB 1090:GOTO 1220
1300 IF A = 18 THEN GOTO 2320
1310 IF A = 20 THEN 2420
1320 IF A = 24 THEN 2260
1330 GOTO 1240:'**** not a legal choice ****
1340 PRINT CHR$(13);:INPUT "Number to dial ";B$
1350 IF B$ = " " THEN RETURN
1360 B$ = 'ATDT' + B$:'**** set up for hayes smartmodems ****
1370 PRINT "Dialing . . . ";
1380 FOR I = 1 TO LEN(B$)
1390 OUT DTA,ASC(MID$(B$,I,1))
1400 TIME1 = 1.6*MHZ:GOSUB 3330
1410 IF I>4 THEN PRINT MID$(B$,I,1);
1420 NEXT
1430 OUT DTA,13
1440 TIME1 = 5*MHZ:GOSUB 3330
1450 A = INP(DTA)
1460 IF (INP(STAT) AND OKI) = OKI THEN 1450
1470 RETURN
1480 PRINT CHR$(13);:LF = 0:CLOSE:INPUT "Name of file to transfer ";FILENAME$
1490 IF FILENAME$ = " " THEN RETURN
1500 RESET:'**** no bdos errors here ****
1509 '**** make sure file name is uppercase ****
1510 FOR I = 1 TO LEN(FILENAME$)
1520 G$ = MID$(FILENAME$,I,1)
1530 IF ASC(G$)>96 AND ASC(G$)<123 THEN G$ = CHR$(ASC(G$)AND 95):MID$(FILENAME$,I,1) = G$
1540 NEXT
1549 '**** a nice touch if you forget ****
1550 IF FILENAME$ = "DIR" THEN FILES:PRINT:GOTO 1480
1560 IF FILENAME$ = "DIR B:" THEN FILES "B:*":PRINT:GOTO 1480
1570 IF FILENAME$ = "DIR A:" THEN FILES "A:*":PRINT:GOTO 1480
1580 OPEN "R",1,FILENAME$
1590 LF = 0
1600 L = LOF(1)

```

```

2240 TIME1 = 6*MHZ:GOSUB 3330:NEXT
2250 PRINT:PRINT "Modem Disconnected ...":GOTO 1220
2259 ***** this copyright must be in the code *****
2260 PRINT:PRINT:PRINT:PRINT CHR$(7);"PX-Transfer Version 1.01 terminating normally ..."
2270 PRINT
2280 PRINT TAB(18);"Copyright (c) 1983"
2290 PRINT TAB(22);"Joseph A. Sabin, Jr."
2300 PRINT TAB(22);"MicroFrame Associates"
2301 ***** because it is giving credit where credit is due *****
2310 GOTO 3320:***** or just end if you like *****
2320 IF DUP = 1 THEN DUP = 0:DUP$ = "Full Duplex type Control R for Half":GOTO 1220
2330 IF DUP = 0 THEN DUP = 1:DUP$ = "Half Duplex type Control R for Full":GOTO 1220
2340 PRINT "— University Computer stopped responding —";CHR$(7)
2350 PRINT "Last Portion sent was ..."
2360 REC = REC - 1
2370 IF REC = 0 THEN PRINT "Nothing was sent, maybe you were not in insert mode":GOTO 1220
2380 GET 1,REC
2390 FIELD 1,128 AS A$
2400 PRINT A$
2410 GOTO 1220
2420 PRINT:PRINT "S to Send — R to Receive file — W if Wrong choice";
2430 A$ = INKEY$:IF A$ = " " THEN 2430 ELSE PRINT A$
2440 IF A$ = "R" OR A$ = "r" THEN GOSUB 2480:GOSUB 2590:GOTO 2630
2450 IF A$ = "S" OR A$ = "s" THEN 1730
2460 IF A$ = "W" OR A$ = "w" THEN 1220
2470 PRINT CHR$(7);:GOTO 2420
2480 PRINT:"Now is the time to insert the disk this file will be written on."
2490 PRINT COMP$;" file name to save input file as";:INPUT FILENAME$
2500 IF FILENAME$ = " " THEN 2420
2510 FOR I = 1 TO LEN(FILENAME$):G$ = MID$(FILENAME$,I,1)
2520 IF ASC(G$)>96 AND ASC(G$)<123 THEN G$ = CHR$(ASC(G$)AND 95):MID$(FILENAME$,I,1) = G$
2530 NEXT
2540 RESET
2550 OPEN "R",1,FILENAME$:L = LOF(1):IF L<>0 THEN PRINT FILENAME$;" Exists — Overwrite (Y/N)?"; ELSE RN = 1:RETURN
2560 A$ = INKEY$:IF A$ = " " THEN 2560 ELSE PRINT A$
2570 IF A$ = "Y" OR A$ = "y" THEN RN = 1 ELSE PRINT:CLOSE:GOTO 2490
2580 RETURN
2590 PRINT:PRINT"Do you want Auto Log Off ?";
2600 A$ = INKEY$:IF A$ = " " THEN 2600 ELSE PRINT A$
2610 IF A$ = "Y" OR A$ = "y" THEN AUTLOG = 1 ELSE AUTLOG = 0
2620 RETURN
2630 PRINT "Soft Carriage Returns are for WordStar(tm) type files":PRINT"Do you want Soft Carriage returns (Y/N)?";
2640 A$ = INKEY$:IF A$ = " " THEN 2640 ELSE PRINT A$
2650 IF A$ = "Y" OR A$ = "y" THEN SOFT = 1 ELSE SOFT = 0
2660 PRINT:PRINT "(1) File name is ";FILENAME$
2670 PRINT "(2) Auto Log Off is ";:IF AUTLOG = 1 THEN PRINT "Active" ELSE PRINT "Inactive"
2680 PRINT "(3) ";:IF SOFT = 1 THEN PRINT "WordStar(tm) type soft carriage returns" ELSE PRINT "Hard carriage returns (STANDARD)"
2690 PRINT "(4) PXEDIT file is open and at TOF:"
2700 PRINT:PRINT"Is this OK (Y/N)? ";
2710 A$ = INKEY$:IF A$ = " " THEN 2710 ELSE PRINT A$
2720 IF A$ = "Y" OR A$ = "y" THEN 2780
2730 IF A$<>"N" AND A$<>"n" THEN 2700 ELSE INPUT "Which one (1,2,3,or 4)";A:IF A<1 OR A>4 THEN 2730
2740 IF A = 1 THEN CLOSE:GOSUB 2480:GOTO 2660
2750 IF A = 2 THEN GOSUB 2590:GOTO 2660
2760 IF A = 3 THEN GOTO 2630
2770 CLOSE:GOTO 1220
2780 T = 1:PRINT:PRINT:LIN = 0:FIELD 1,128 AS A$
2790 PRINT"— TYPE CONTROL T TO TERMINATE RECEIVE —"
2799 ***** replace this — ↓ — with whatever character gives the next line of text from the mainframe editor *****
2800 LIN = LIN + 1:OUT DTA,ASC("N"):PRINT CHR$(13);"Receiving Line #";LIN;"Current file size = ";MID$(STR$(INT((RN + T - 1)/8 + 1)),2);"K";
2810 IF INKEY$ = CHR$(20) THEN 3010:***** bail out of receive *****
2820 TIME1 = 1.6*MHZ:GOSUB 3330
2830 OUT DTA,13
2838 ***** this next part determines when the EOF: occurs on the mainframe *****
2839 ***** this is EOF: start ↓ is normal continue — this ↓ is E as in EOF: so go end file *****
2840 A = (INP(DTA) AND &H7F):IF A = 95 THEN 2850 ELSE C = INP(STAT):IF A = 69 THEN 3010 ELSE 2840
2850 IF (INP(STAT) AND OKI) = OKI THEN A = (INP(DTA) AND &H7F) ELSE 2850
2860 IF A = 13 THEN 2890
2870 IF A<31 THEN 2850
2880 TEMP$ = TEMP$ + CHR$(A):GOTO 2850
2890 IF SOFT = 0 THEN TEMP$ = TEMP$ + CHR$(13) + CHR$(10) ELSE TEMP$ = TEMP$ + CHR$(141) + CHR$(10)

```

```

1610 IF L = 0 AND LF = 0 THEN KILL FILENAME$:PRINT FILENAME$;" is empty":PRINT"Try again":GOTO 1480
1620 LF = LF + L
1630 IF L = 128 THEN GET 1,LF + 1:GOTO 1600
1640 PRINT FILENAME$;" is ";LF;"records long"
1650 PRINT STRING$(50," - ")
1660 PRINT "| Transfer time is approx. = ";INT(128*LF/(21)/60);"Minutes";INT(128*LF/21) - INT(128*LF/(21)/60)*60;"Seconds |"
1670 PRINT STRING$(50," - ")
1680 PRINT"Be sure that the University computer is ready."
1690 PRINT STRING$(50," - ")
1700 PRINT"| This means in XEDIT and in the insert mode. |"
1710 PRINT STRING$(50," - ")
1720 RETURN
1730 PRINT:GO SUB 1480
1740 IF FILENAME$ = " " THEN GOTO 2420
1750 PRINT"Do you want Auto Log Off ?";
1760 A$ = INKEY$:IF A$ = " " THEN 1760 ELSE PRINT A$
1770 IF A$ = "Y" OR A$ = "y" THEN AUTLOG = 2 ELSE AUTLOG = 0
1780 PRINT:PRINT "(1) File name is ";FILENAME$
1790 PRINT "(2) Auto Log Off is ";:IF AUTLOG = 2 THEN PRINT "Active" ELSE PRINT "Inactive"
1800 PRINT "(3) PXEDIT is in the Insert Mode "
1810 PRINT:PRINT"Is this OK (Y/N)? ";
1820 A$ = INKEY$:IF A$ = " " THEN 1820 ELSE PRINT A$
1830 IF A$ = "Y" OR A$ = "y" THEN 1870
1840 IF A$ <> "n" AND A$ <> "N" THEN 1810 ELSE INPUT"Which one (1,2,or 3)";A:IF A < 1 OR A > 3 THEN 1840
1850 IF A = 1 THEN GOSUB 1480:GOTO 1780 ELSE IF A = 2 THEN GOTO 1750
1860 CLOSE:GOTO 1220
1870 REC = 1:CHECK = 1
1880 GET 1,REC
1890 FIELD 1,128 AS SEND$
1900 SENT = 1
1910 C$ = CHR$(ASC(MID$(SEND$,SENT,1)) AND &H7F):IF C$ = CHR$(26) AND REC = LF THEN 2150
1920 IF CHECK = 0 THEN CHECK = 1
1930 IF C$ = "#" THEN SENT = SENT + 1:IF SENT < 129 THEN 1910 ELSE 1990
1940 IF INKEY$ = CHR$(24) THEN 2260:**** this is Control X to stop transfer ****
1950 IF ASC(C$) < 31 AND C$ <> CHR$(13) THEN SENT = SENT + 1:IF SENT < 129 THEN 1910 ELSE 1990
1960 PRINT C$;:IF C$ <> CHR$(13) THEN OUT DTA,ASC(C$):TIME1 = 0.6*MHZ:GOSUB 3330
1970 IF C$ = CHR$(13) THEN GOTO 2020
1980 IF CR = LL THEN PRINT CHR$(13);:GOTO 2020
1990 IF SENT >= 128 THEN REC = REC + 1:ELSE SENT = SENT + 1:CHECK = CHECK + 1:CR = CR + 1:GOTO 1910
2000 IF REC > LF THEN GOTO 2150
2010 GOTO 1880
2020 PRINT CHR$(10);:CR = 1:TEST = 1:IF CHECK <> 1 THEN OUT DTA,13
2030 IF CHECK = 1 THEN CHECK = 0:GOTO 1990
2040 FOR TEST = 1 TO 240*MHZ
2050 A = (INP(DTA) AND &H7F)
2059 *** ↓ set the number below equal to the incoming prompt to look for ****
2060 IF A = 7 THEN CHECK = 0:TIME1 = 0.4*MHZ:GOSUB 3330:GOTO 1990
2070 NEXT
2080 TEST = 1
2090 OUT DTA,13
2100 A = (INP(DTA) AND &H7F)
2109 *** ↓ the same prompt as 2060 ***
2110 IF A = 7 THEN 2120 ELSE TEST = TEST + 1:IF TEST = 3000 THEN 2340 ELSE 2110
2119 ***** this is an attempt to put the ↓ system back in insert mode *****
2120 TIME1 = 0.4*MHZ:GOSUB 3330:OUT DTA,ASC("("):TIME1 = 6*MHZ:GOSUB 3330:OUT DTA,13
2129 ***** again one more time -v- check for that prompt *****
2130 A = (INP(DTA) AND &H7F):IF A = 7 THEN 2140 ELSE 2130
2140 TIME1 = 0.4*MHZ:GOSUB 3330:CHECK = 0:GOTO 1990
2150 CLOSE:FILENAME$ = ":PRINT:PRINT "Transfer complete":PRINT"log off . . .":OUT DTA,13:IF AUTLOG = 2 THEN 3110 ELSE 1220
2160 A = INP(DTA):A = INP(DTA):TIME1 = 5*MHZ:GOSUB 3330
2170 GOTO 1220
2179 **** this is to get the modem's attention ****
2180 DISC$(1) = "+++":DISC$(2) = "ATH" + CHR$(13)
2181 ****and this ↑ is to hang up the modem ****
2190 FOR J = 1 TO 2:FOR I = 1 TO LEN(DISC$(J))
2200 OUT DTA,ASC(MID$(DISC$,I,1))
2209 **** ↓ pause between char. ↓ wait for a second for modem ****
2210 TIME1 = 26*MHZ:GOSUB 3330:NEXT I:TIME1 = 50*MHZ:GOSUB 3330:NEXT
2219 ***** this is to avoid the garbage from the modem *****
2220 FOR I = 1 TO 6
2230 A = INP(DTA)

```

```

2900 A=(INP(DTA) AND &H7F)
2910 IF A = 17 THEN 2920 ELSE 2900
2920 LA = LEN(A$(T)):LT = LEN(TEMP$)
2930 IF LA + LT > 128 THEN A$(T) = A$(T) + LEFT$(TEMP$,128-LA):T = T + 1: A$(T) = RIGHT$(TEMP$,LT - (128-LA)):TEMP$ = "":ELSE
    A$(T) = A$(T) + TEMP$:TEMP$ = "":GOTO 2800
2940 IF T = 5 THEN 2950 ELSE 2800
2950 FOR I = 1 TO 4
2960 LSET A$ = A$(I)
2970 PUT 1,RN
2980 RN = RN + 1
2990 NEXT
3000 A$(1) = A$(5):FOR I = 2 TO 5:A$(I) = "":NEXT:I = 1:GOTO 2800
3009 ***** CP/M EOF          ↓          ↓ number of current buffer strings
3010 A$(T) = A$(T) + CHR$(26):FOR I = 1 TO T
3020 LSET A$ = A$(I)
3030 PUT 1,RN
3040 RN = RN + 1
3050 NEXT
3060 CLOSE
3070 PRINT:PRINT"File named ";FILENAME$;" has been received . . ."
3080 IF AUTLOG = 1 THEN 3180 ELSE 1220
3100 ***** AUTO LOG OFF ROUTINE *****
3110 OUT DTA,13:TIME1 = *MHZ:GOSUB 3330
3120 A=(INP(DTA) AND &H7F)
3129 ***      ↓ wait for this character ***
3130 IF A = 8 THEN 3140 ELSE 3120
3139 **** -v- command to mainframe to save transferred file ****
3140 A$ = "SAVE" + CHR$(13):TIME1 = 6*MHZ:GOSUB 3330
3150 FOR I = 1 TO LEN(A$)
3160 OUT DTA,ASC(MID$(A$,I,1))
3170 TIME1 = 6*MHZ:GOSUB 3330:NEXT:GOTO 3190
3180 OUT DTA,13
3189 ***** this part works both send and receive *****
3190 A=(INP(DTA) AND &H7F)
3199 ***      ↓ wait for this character **
3200 IF A = 8 THEN 3210 ELSE 3190
3209 ****      ↓ command to mainframe to quit editor ****
3210 A$ = "QUIT" + CHR$(13):TIME1 = 6*MHZ:GOSUB 3330
3220 FOR I = 1 TO LEN(A$)
3230 OUT DTA,ASC(MID$(A$,I,1))
3240 TIME1 = 6*MHZ:GOSUB 3330:NEXT
3250 TIME1 = 6*MHZ:GOSUB 3330
3260 A=(INP(DTA) AND &H7F)
3269 ***      wait for this character ***
3270 IF A = 8 THEN 3280 ELSE 3260
3279 ***      ↓ magic word to exit from the time sharing ***
3280 A$ = "LOG" + CHR$(13):TIME1 = 6*MHZ:GOSUB 3330
3290 FOR I = 1 TO LEN(A$)
3300 OUT DTA,ASC(MID$(A$,I,1))
3310 TIME1 = 6*MHZ:GOSUB 3330:NEXT
3320 END
3329 ***** delay loop *****
3330 FOR TIME = 1 TO TIME1:NEXT:RETURN

```

